

## Description

Some DSGE models capture delayed effects by including a second-order lag of a control variable and excluding the first-order lag. The second-order lag is a problematic term that does not fit into the form required to solve a structural model for its state-space form. This entry shows how to solve this problem by defining new state variables and rewriting the equations.

## Remarks and examples

Remarks are presented under the following headings:

*The model*

*Parameter estimation*

### The model

Consider a model in which changes in hours worked take two periods to adjust because next period's hours have already been budgeted. In this model, the second-order lag of changes in hours worked is included, and the first-order lag is excluded. Equations (1)–(4) specify such a model of growth in hours worked and of consumption growth.

$$n_t = b_1 n_{t-2} + w_t - \gamma c_t \quad (1)$$

$$c_t = (1 - h)w_t + hE_t c_{t+1} + r_t \quad (2)$$

$$w_{t+1} = \rho w_t + \xi_{t+1} \quad (3)$$

$$r_{t+1} = \epsilon_{t+1} \quad (4)$$

Equation (1) specifies that the growth rate of hours worked  $n_t$  depends on a second-order lag of itself, wage growth  $w_t$ , and consumption growth  $c_t$ . Equation (2) specifies that consumption growth is a linear combination of wage growth, expected future consumption growth  $E_t c_{t+1}$ , and the interest rate  $r_t$ . Equation (3) specifies an autoregressive process for wage growth. Equation (4) specifies that interest rate is just a shock. The control variables are  $n_t$  and  $c_t$ . The state variables are  $w_t$  and  $r_t$ .

One cannot solve the model in (1)–(4) for the state-space form because the problematic term  $b_1 n_{t-2}$  does not fit into the required form. To accommodate this term, we define two new state variables, one for  $n_{t-1}$  and one for  $n_{t-2}$ . We define new state variables instead of new control variables because lags of the control are predetermined and thus exogenous. The model with new state variables is

$$n_t = b_1 L2n_t + w_t - \gamma c_t \quad (5)$$

$$c_t = (1 - h)w_t + hE_t c_{t+1} + r_t \quad (6)$$

$$w_{t+1} = \rho w_t + \xi_{t+1} \quad (7)$$

$$r_{t+1} = \epsilon_{t+1} \quad (8)$$

$$Ln_{t+1} = n_t \quad (9)$$

$$L2n_{t+1} = Ln_t \quad (10)$$

Equation (9) defines the new state for  $n_{t-1}$ , and (10) defines  $L2n_t$  to be the new state for  $n_{t-2}$ . The  $L2n_t$  in (5) replaces  $n_{t-2}$  in (1).

## Parameter estimation

We specify  $n$  and  $c$  as observed control equations. We specify  $w$ ,  $r$ ,  $Ln$ , and  $L2n$  as state equations. We specify that  $w$  and  $r$  are subject to shocks; the new states to accommodate  $n_{t-2}$  are not subject to shocks.

```
. use https://www.stata-press.com/data/r19/usmacro2
(Federal Reserve Economic Data - St. Louis Fed, 2017-01-15)

. dsge (n = {b1}*L2n + w - {gamma}*c)
>      (c = (1-{h})*w + {h}*F.c + r)
>      (F.w = {rho}*w, state)
>      (F.r = , state)
>      (F.L2n = Ln, state noshock)
>      (F.Ln = n, state noshock)
(setting technique to bfgs)
Iteration 0:  Log likelihood = -2325.1996
Iteration 1:  Log likelihood = -1277.0146   (backed up)
Iteration 2:  Log likelihood = -1193.4512   (backed up)
Iteration 3:  Log likelihood = -1189.3181   (backed up)
Iteration 4:  Log likelihood = -1188.2629   (backed up)
(switching technique to nr)
Iteration 5:  Log likelihood = -1187.9872   (backed up)
Iteration 6:  Log likelihood = -1147.3696
Iteration 7:  Log likelihood = -1131.4924
Iteration 8:  Log likelihood = -1129.035
Iteration 9:  Log likelihood = -1129.0181
Iteration 10: Log likelihood = -1129.0181

DSGE model

Sample: 1955q1 thru 2015q4                      Number of obs = 244
Log likelihood = -1129.0181
```

	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
/structural						
b1	.1320846	.0608727	2.17	0.030	.0127763	.2513928
gamma	.3609253	.1298388	2.78	0.005	.1064459	.6154048
h	.7238121	.0406724	17.80	0.000	.6440958	.8035285
rho	.6177969	.0533568	11.58	0.000	.5132195	.7223743
sd(e.w)	3.0338	.2423858			2.558733	3.508868
sd(e.r)	1.970288	.1574528			1.661686	2.27889

Looking at the confidence interval for  $b1$ , we conclude that the second-order lag of hours' growth impacts current hours' growth.

## Also see

[DSGE] [Intro 2](#) — Learning the syntax

[DSGE] [Intro 4](#) — Writing a DSGE in a solvable form

Stata, Stata Press, Mata, NetCourse, and NetCourseNow are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow is a trademark of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).